
Einleitung 

Das intelligente Haus oder die intelligente Wohnung ist der Traum von Mietern und 
selbst nutzenden Eigentümern gleichermaßen: Eine fernsteuerbare Heizungssteue-
rung hilft, Energiekosten zu sparen, schafft aber gleichzeitig die Möglichkeit, nach dem 
Nachmittag auf der Loipe in ein warmes Badezimmer zu kommen. Eine geschickt 
umgesetzte Einlasssteuerung erlaubt es, Schlüssel zu Hause zu lassen und sich statt-
dessen per PIN-Code oder mit dem Smartphone zu authentifizieren.  

In einer idealen Welt greifen alle automatisierbaren Komponenten ineinander und 
beziehen externe Dienste mit ein: Da arbeiten Wettervorhersagen aus dem Internet 
mit lokalen Messergebnissen zusammen, um rechtzeitig die Heizung vom Niedertem-
peraturbetrieb auf Winter umzuschalten. Umgekehrt arbeiten im Sommer Klima-
anlage und Beschattung so zusammen, dass bereits vor dem kritischen Sonnenstand 
die Wohnung abgeschattet und klimatisiert ist – bei möglichst geringem Energiever-
brauch. Eine automatische Erkennung der Anwesenden (an Smartphone oder einem 
NFC-Chip am Schlüsselbrett erkannt) sorgt dafür, dass Teile oder die gesamte Woh-
nung nur dann beheizt und beleuchtet sind, wenn alle Bewohner auch anwesend sind.  

Doch leider existiert diese ideale Welt nicht: Erschwingliche Hausautomationslösun-
gen zum Nachrüsten sind Produkte der letzten fünf Jahre. Meist decken sie nur Teil-
aspekte ab, und oft sind sie nicht besonders gut auf einzelne Märkte zugeschnitten: So 
wirbt ein großer Hersteller von Netzwerkequipment für den häuslichen Bereich mit 
Grafiken von Heizkörpern – obwohl der Hersteller keine ferngesteuerten Heizkörper-
thermostate anbietet. Tatsächlich möchte er seine Funksteckdosen verkaufen. Das 
funktioniert in Norwegen, wo fast 100 % des elektrischen Stroms durch Wasserkraft 
erzeugt werden (und daher Strom ohne schlechtes ökologisches Gewissen zum 
Heizen genutzt werden kann), nicht jedoch in Deutschland, wo aufgrund des hohen 
Anteils fossiler Energieträger zur Stromversorgung die Warmwasser-Zentralheizung 
noch dominiert. Wer eine erschwingliche Hausautomation nachrüsten möchte, sieht 
sich also mit dem Problem konfrontiert, verschiedene Systeme zur Zusammenarbeit 
bewegen zu müssen. Dazu kommt der Faktor Zeit: Da eine nachgerüstete Hausauto-
mation »organisch« den eigenen Bedürfnissen folgt, werden Sie im Zuge Ihrer Arbeit 
manche Systeme fallen lassen, andere werden neu hinzukommen.  

Weitere Aspekte sind Sicherheit und Datenschutz: Viele integrierte Systeme sind als 
Blackbox ausgelegt, teilweise werden Ihre Daten »in der Cloud« vorgehalten. Am 
eklatantesten ins Bewusstsein drang dieser Aspekt mit der Übernahme des 
Thermostatherstellers NEST durch Google im Februar 2014: Künftig werden Googles 
Algorithmen für recht präzise Vorhersagen, wann wer zu Hause sein wird, herangezo-
gen – solche selbst lernenden Systeme sind gut für den Energieverbrauch und verspre-
chen viel Komfort, wecken aber gleichzeitig neue Begehrlichkeiten für NSA & Co.  

Die Motivation für dieses Buch war daher, auf Basis der Arduino-Familie die Kontrolle 
über das eigene Heim wiederzuerlangen. Arduino deshalb, weil diese Mikrocontroller-
Plattform vom 3 Euro billigen Minimalcontroller bis hin zu vollwertigen Linux-Boards 
mit Webserver, WLAN und MicroSD-Slot alles zum Regeln, Steuern und Messen mit-
bringt. Dabei sind alle Mitglieder der Familie nach ein und demselben Konzept über 
die komfortable Entwicklungsumgebung zu programmieren. Ausflüge in andere Pro-
grammiersprachen als Arduinos einfach zu erlernenden C-Dialekt sind selten nötig. 
Dabei wollen wir es nicht bei der Integration vorgefertigter Komponenten belassen: 
Was es nicht gibt oder zu teuer ist, wird selbst gebaut. 

Großer Projektteil 

3.1 Kommunikation mit Funkaktoren 

Projekt: Arduino schaltet Funksteckdosen 
In diesem Projekt verwenden wir einen einfachen RF-Link-Sender, um Funksteckdosen 
an- und auszuschalten. Ziel ist es, eine verlässliche Einwegkommunikation aufzu-
bauen, die Adressierung korrekt zu erreichen und die Reichweite des Senders zu 
testen.  

Benötigte Bauteile 

] beliebige Arduino-Platine (bevorzugt Uno oder Pro Mini) 

] 434-MHz-RF-Link-Sender (2 bis 4 Euro) 

] Funksteckdose 434 MHz (2er- oder 3er-Pack 10 bis 25 Euro) 

] 18 cm Schaltdraht YV 

Benötigte Software 

] RCSwitch-Bibliothek – getestet mit Version 2.51 (code.google.com/p/rc-switch/) 

Die theoretischen Grundlagen des bei billigen Funksteckdosen verwendeten Protokolls 
haben wir bereits im letzten Kapitel beschrieben: einfache Amplitudenmodulation, 

geringer Energieverbrauch beim Senden, aber keine Bestätigung, ob ein Befehl ange-
kommen ist und keinerlei Sicherheit. Dennoch gibt es viele Bereiche im Haushalt, in 
denen diese Steckdosen ideal sind: sei es bei der Beleuchtung des Weihnachtsbaums 
oder um einfach sicherzustellen, dass bestimmte Verbraucher – beispielsweise Stand-
by-Stromfresser – zu bestimmten Zeiten garantiert ausgeschaltet sind.  

Kommen wir zur Praxis: Funksteckdosen verwenden zwei verschiedene Verfahren, 
den Haus- oder Systemcode sowie den Gerätecode zu setzen: Einige Modelle nutzen 
zwei simple Drehschalter mit jeweils vier Positionen zum Setzen der jeweiligen Codes. 
Das macht im Ergebnis maximal vier Hauscodes mit jeweils maximal vier Steckdosen. 
Andere Geräte verwenden je fünf Dip-Schalter für Hauscode und Geräteadresse und 
decken folglich mit 32 Hauscodes à 32 Geräten weit mehr ab. Beim Neukauf sollten 
Sie schon wegen der geringeren Gefahr von Kollisionen in der Nachbarschaft Geräten 
mit Dip-Schaltern den Vorzug geben.  

 

Bild 3.1: Zwei verschiedene Codierungsarten: links Dip-Switches, rechts Drehschalter – 
RCSwitch kann beide ansprechen. 

Zur Ansteuerung gibt es zwei Möglichkeiten. Die wohl simpelste ist die digitale 
Ansteuerung des Sendebausteins SC5262 (oder seiner Verwandten) vom Arduino 
aus. Dazu benötigen Sie besagten IC und einen 434-MHz-RF-Link-Sender. An die 
Bauteile kommen Sie heran, indem Sie eine Fernbedienung zerlegen. Die Alternative 
ist, den Arduino die Aufbereitung des Signals erledigen zu lassen und dieses dann zum 
434-MHz-RF-Link-Sender zu schicken. Das spart ein paar Cent oder das Schlachten 
einer Fernbedienung, hat aber vor allem den Vorteil der viel größeren Flexibilität: Sie 
können nicht nur die beiden Typen von Schaltsteckdosen ansteuern, sondern das 
simple Protokoll auch für eigene Zwecke erweitern – Möglichkeiten, die die fest ver-
drahtete Logik der Sende-ICs nicht bietet.  

Als Bibliothek auf dem Arduino bietet sich hierfür RCSwitch an. Ein Vorteil dieser 
Bibliothek ist, dass Sie selbst nicht die Stellung der Dip-Switches und Positionsschalter 
in die entsprechenden Bitfolgen übersetzen müssen: RCSwitch stellt die Funktionen 

switchOn() und switchOff() zur Verfügung, die zwei Parameter erwarten: Hauscode 
und Geräteadresse – als Integer 1 bis 4 (Drehschalter) oder als String der Form 01100 
bei Dip-Schaltern. Importieren Sie die Bibliothek RCSwitch in Ihre Arduino-Entwick-
lungsumgebung (Hinweise zum Import von Bibliotheken im Anhang) und erstellen Sie 
folgenden Sketch (Hauscode sei 2, Steckdosencode 1):  

#include <RCSwitch.h> 
 
RCSwitch mySwitch = RCSwitch(); 
int ledPin = 13; 
 
void setup() { 
  mySwitch.enableTransmit(10); 
  pinMode(ledPin, OUTPUT); 
} 
 
void loop() { 
  mySwitch.switchOn(2, 1); 
  digitalWrite(ledPin, HIGH); 
  delay(2000);   
  mySwitch.switchOff(2, 1); 
  digitalWrite(ledPin, LOW); 
  delay(2000);   
} 

Sollten Ihre Steckdosen Dip-Switches verwenden, sind switchOn() und switchOff 
etwas anders aufzurufen – die Steckdosen mit den Dip-Schalterstellungen 00001, 
00010, 00100, 01000 und 10000 sind dabei von der Fernbedienung erreichbar, Steck-
dosen, bei denen mehr als ein Dip-Schalter auf On steht, nur per Arduino:  

mySwitch.switchOn("00110", "00010"); 

Verbinden Sie nun Arduino und 434-MHz-RF-Link-Sender, wie in der Schaltskizze 
gezeigt: Pin 10 des Arduino auf DATA des Sendemoduls. Verlassen Sie sich dabei nur 
auf die Beschriftungen des Sendemoduls, nicht auf die Schaltskizze – es ist ein halbes 
Dutzend verschiedener Bauformen des kleinen Sendemoduls auf dem Markt. Als 
Antenne dient der starre Schaltdraht, der vom Fußpunkt aus ca. 17,3 cm lang sein 
sollte. Als Faustformel für die Antennenlänge gilt ein Viertel der Wellenlänge  – diese 
beträgt bei 434 MHz ca. 69 cm. Reicht der Platz nicht für eine stehende Antenne, 
drehen Sie den Schaltdraht über einen Bleistift wie über einen Lockenwickler.  

Bild 3.2: Der mit DATA beschriftete Anschluss des Senders wird mit Pin 10 des Arduino 
verbunden. 

Bild 3.3: Vergessen Sie die Antenne nicht: Hier ragt ein Schaltdraht senkrecht nach oben, 
wichtiger als die Ausrichtung ist die richtige Länge von einem Viertel der Wellenlänge. 

Reichweite testen 

Mit der stehenden 17,3-cm-Antenne sollten sich bei guten Bedingungen auch in 
Wohnungen etwa 15 m überbrücken lassen. Stehen Wände aus Stahlbeton im Weg 
oder sind Störsender wie ein Babyfon oder des Nachbars Funkkopfhörer aktiv, sinken 
Reichweite und Zuverlässigkeit. Sie können folgenden Sketch verwenden, um bei 
Steckdosen mit Positionsschalter einfach alle Steckdosen rhythmisch an- und abzu-
schalten und so die beste Position des Senders zu ermitteln. Bei Steckdosen mit Dip-
Swich macht es die schiere Menge möglicher Konstellationen unsinnig, alle durchzu-
probieren – packen Sie bei Ihnen vorhandene Steckdosen in ein Array, über das Sie 
iterieren.  

#include <RCSwitch.h> 
 
RCSwitch mySwitch = RCSwitch(); 
 
void setup() { 
  mySwitch.enableTransmit(10); 
} 
 
void loop() { 
  for (int h = 1; h < 5; h++) { 
    for (int o = 1; o < 5; o++) {  
      mySwitch.switchOn(h, o); 
      delay(1000); 
    } 
    for (int o = 1; o < 5; o++) {  
      mySwitch.switchOff(h, o); 
      delay(1000); 
    } 
  } 
} 

Ist die Zuverlässigkeit der Funkübertragung schlecht, senden Sie die Kommandos im 
Abstand von ein paar Hundert Millisekunden einfach zwei- oder dreifach. Genügt auch 
das nicht, bleibt die Möglichkeit, den Arduino über eine 12-V-Spannungsquelle zu ver-
sorgen und diese auch das Funkmodul speisen zu lassen – die meisten einfachen 
Funkmodule sind von 3 V bis 12 V spezifiziert.  

Diesen Sketch sollten Sie nicht länger als eine Minute ausprobieren und schon gar 
nicht auf einem laufenden Arduino mit Sendemodul vergessen! Da alle Steckdosen auf 
allen Hauscodes durchprobiert werden, besteht die Gefahr, dass es auch beim Nach-
barn blinkt. Zudem gebietet die Fairness gegenüber anderen Funkanwendern, die 
433,92-MHz-Punktfrequenz nicht über Gebühr zu beanspruchen und damit für 
andere legitime Nutzer zu blockieren. 

Projekt: Erweiterung zur Schaltuhr 
Dieses Projekt kombiniert den Funksender aus dem letzten Projekt mit einer Uhr – 
Real-Time-Clock- oder Funkuhrempfänger sind möglich – zu einer Schaltuhr für Funk-
steckdosen.  

Benötigte Bauteile 

] alle Bauteile des letzten Projekts 

] ein RTC-Baustein mit Knopfzelle, beispielsweise DS3231 (Banggood SKU082052) 
oder DS1307, ca. 3 bis 10 Euro 

] alternativ: DCF77-Empfänger (Conrad 641138 oder Pollin 810054), ca. 5 bis 15 
Euro 

Benötigte Software 

] DCF77-Bibliothek: github.com/thijse/Arduino-Libraries/downloads 

] DS1307RTC-Bibliothek: www.arduino.cc/playground/Code/Time 

] Time-Bibliothek: www.arduino.cc/playground/Code/Time 

Zeitsignal: Real Time Clock 

Da der Arduino keine interne Uhr besitzt, die bei einer Stromunterbrechung weiter-
laufen kann, muss eine Uhr angeschlossen werden. Bewährt haben sich RTC-Module, 
die von einer Knopfzelle gespeist werden. Wir haben uns für einen DS3231-Baustein 
entschieden, dieser nutzt wie DS1307 die I²C-Pins A4 und A5, ist aber für eine 
breitere Eingangsspannung spezifiziert und deutlich genauer: Der Hersteller garantiert 
weniger als zwei Minuten Drift pro Jahr, was auch durch eine Temperatur-
kompensation erreicht wird. Beim DS1307 dagegen sind eher zwei Minuten pro Monat 
üblich. Ein netter Nebeneffekt ist, dass der Temperatursensor direkt über I²C ausge-
lesen werden kann und so in vielen Anwendungsfällen ein separater Thermistor 
(wärmeabhängiger Widerstand) unnötig ist. Beide Bausteine werden von der 
DS1307RTC-Bibliothek angesteuert, und auch wenn so manches Suchergebnis im 
Web komplett unterschiedliche Bausteine nahelegt: Mittlerweile wurde die Unter-
stützung für DS3231 vollständig in die Bibliothek aufgenommen. Schließen Sie das 
RTC-Modul wie in der Skizze gezeigt an: A4 des Arduino nach SDA des Moduls und 
A5 auf SDC. Die Versorgungsspannung VCC darf beim DS3231 5 V betragen, einige 
mit DS1307 bestückte Platinen setzen 3,3 V voraus.  

Bild 3.4: Das DS3231-RTC-Modul wird per I²C mit dem Arduino verbunden. 

Laden Sie nun den folgenden Sketch auf den Arduino. Im loop() initialisiert er ein 
leeres Objekt des Typs tmElements_t, in das die Funktion RTC.read() regelmäßig 
Zeitinformationen einliest. Die Ausgabe erfolgt schließlich über die serielle Konsole.  

#include <DS1307RTC.h> 
#include <Time.h> 
#include <Wire.h> 
 
void setup() { 
  Serial.begin(9600); 
} 
 
void loop() { 
  tmElements_t tm; 
  if (RTC.read(tm)) { 
    print2digits(tm.Hour); Serial.write(':'); 
    print2digits(tm.Minute); Serial.write(':'); 
    print2digits(tm.Second); Serial.print(", Datum: "); 
    Serial.print(tm.Day); Serial.write('.'); 
    Serial.print(tm.Month); Serial.write('.'); 

    Serial.print(tmYearToCalendar(tm.Year)); 
    Serial.println(); 
  }  
  delay(1000); 
} 
 
void print2digits(int number) { 
  if (number >= 0 && number < 10) { 
    Serial.write('0'); 
  } 
  Serial.print(number); 
} 

In den seltensten Fällen wird die RTC ab Werk richtig gesetzt sein, Sie müssen folglich 
erst einmal an der Uhr drehen. Die dafür bereitstehenden Funktionen der DS1307-
Bibliothek sind simpel, so ist es mit wenig Aufwand möglich, einen String mit der 
aktuellen Uhrzeit zum an der seriellen Schnittstelle angeschlossenen Arduino zu 
schicken und dort zu verarbeiten. Eine automatische Übernahme der Systemzeit mit 
dieser Methode ist aber nur mit der Linux-Shell möglich. Wir fanden daher das Bei-
spiel SetTime der Bibliothek DS130RTC sehr sympathisch: Es liest den Zeitstempel des 
Compilers aus dem hochgeladenen Sketch aus und setzt die Uhr auf diesen Zeitpunkt. 
Das bedeutet natürlich zunächst, dass die Systemzeit des PCs mit der Arduino-
Entwicklungsumgebung stimmen muss und dass ein von diesem Sketch erstelltes 
Binary nur wenige Sekunden nach dem Kompilieren einigermaßen präzise ist. Zudem 
ist nicht immer absehbar, wie lange der Upload und der Reset des Arduino dauert – 
wenige Sekunden Abweichung sind jedoch für die meisten Zwecke zu verschmerzen. 
Auf dem Testsystem betrug die Abweichung 12 bis 14 Sekunden – wenn Sie es wün-
schen, können Sie die Funktion parse() um eine harte Abweichung anpassen. 
Resetten Sie nach dem Upload von SetTime den Arduino keinesfalls und trennen Sie 
ihn auch nicht vom PC, sondern ersetzen Sie SetTime sofort wieder durch unser 
Beispiel zum Auslesen der Uhr. 

Nun können Sie damit beginnen, RTC und Funksender zusammenzubringen. Für die 
Speicherung der An- und Abschaltzeiten haben wir uns für ein simples zweidimensio-
nales Array entschieden. Dessen Größe und damit die Zahl der Schaltvorgänge ist fix. 
Zur Vereinfachung und besseren Übersicht sind genau drei Steckdosen mit je zwei 
An- und zwei Abschaltvorgängen pro Tag vorgesehen, diese müssen zunächst auf 
vollen Stunden liegen – nur einen Aus/An-Zyklus erreichen Sie, indem Sie zwei 
Anschaltzeiten vor eine Abschaltzeit legen. Relativ einfach gestaltet sich die Entschei-
dung, wann geschaltet werden soll: Ist eine volle Stunde erreicht, werden beide Arrays 
durchlaufen, und gegebenenfalls wird ein Schaltvorgang ausgelöst. Das Beispiel geht 
implizit von einem Hauscode h und den Gerätecodes 1 bis 3 aus. Bei Steckdosen mit 
Dip-Switches müssen Sie deren Stellungen in einem eigenen Array dipCode notieren 
und über switchOn(h, dipCode[i]) schalten.  

int onMatrix[3][2] = { 
  { 5, 14 }, 
  { 11, 19 }, 

  { 10, 11 } 
}; 
int offMatrix[3][2] = { 
  { 3, 12 }, 
  { 12, 20 }, 
  { 3, 19 } 
}; 
 
// zur vollen Stunde 
void traverseMatrix(int hour) { 
  for (int i=0; i<3; i++) { 
    for (int j=0; j<2; j++) { 
      if (onMatrix[i][j] == hour) { 
        mySwitch.switchOn(h, i+1); 
        delay(500); 
      } 
    } 
  } 
} 

Bleibt die Frage, wie beim Start des Mikrocontrollers zum Beispiel nach einem Strom-
ausfall oder bei der Programmierung mit einem anderen Schaltmuster zu verfahren ist: 
In einem ersten Sketch habe ich mich für die allereinfachste Methode entschieden und 
ging davon aus, dass zum Zeitpunkt des Starts alle Steckdosen richtig geschaltet sind. 
Mangels Rückkanal der 434-MHz-Steckdosen ist das eine sehr unbefriedigende 
Annahme, denn:  

] War Stromausfall, sind alle 434-MHz-Steckdosen aus, denn die Relais waren 
einen Moment ohne Stromversorgung. 

] Wurde nur der Arduino kurz von der Stromversorgung getrennt, sind im Regelfall 
keine Aktionen nötig. 

] Haben Sie den Arduino mit einem anderen Schaltmuster versehen, haben einige 
Verbraucher nicht ihren Sollstatus. 

Bild 3.5: Der Anschluss der Funksteckdosenschaltuhr kombiniert RF-Link-Sender und RTC, alle 
Pin-Belegungen bleiben gleich. 

Die einfachste Variante wäre, beim Start des Mikrocontrollers alle Stunden von 0 bis 
23 Uhr und dann erneut von 0 Uhr bis zur aktuellen Stunde einmal unter Aufruf der 
Funktion traverseMatrix() zu durchzulaufen. Wir haben schließlich die nächstauf-
wendigere Methode gewählt und laufen einmal von 0 bis 23 Uhr und dann noch ein-
mal von 0 Uhr bis zur aktuellen Zeit durch. Sinn dieses mehrfachen Durchlaufs ist es, 
sicherzustellen, dass bei einem Reboot des Mikrocontrollers um 1 Uhr nachts die letzte 
Schaltaktion, die vielleicht um 22:00 stattgefunden hat, berücksichtigt wird – das 
schlägt sich in der Funktion setStartState() nieder und resultiert im folgenden 
Sketch:  

#include <DS1307RTC.h> 
#include <Time.h> 
#include <Wire.h> 
#include <RCSwitch.h> 
 
const int h = 1; 
RCSwitch mySwitch = RCSwitch(); 
int onMatrix[3][2] = { 
  { 10, 17 }, 
  { 11, 18 }, 
  { 10, 11 } 
}; 
int offMatrix[3][2] = { 
  { 11, 22 }, 
  { 12, 23 }, 
  { 3, 19 } 
}; 
int savedState[3] = { 0, 0, 0 }; 
int startState[3] = { 0, 0, 0 }; 
 
void setStartState(int hour) { 
  for (int k=0; k<=24+hour; k++) { 
    for (int i=0; i<3; i++) { 
      for (int j=0; j<2; j++) { 
        if (onMatrix[i][j] == k % 24) { 
          startState[i] = 1;  
        } 
        if (offMatrix[i][j] == k % 24) { 
          startState[i] = 0;  
        } 
      } 
    } 
  } 
  for (int i=0; i<3; i++) { 
     if (startState[i] > 0) { mySwitch.switchOn(h, i+1); } 
     if (startState[i] < 1) { mySwitch.switchOff(h, i+1); } 
     delay(100);  
  } 



3.1  Kommunikation mit Funkaktoren  
 

 91 

} 
 
void traverseMatrix(int hour) { 
  for (int i=0; i<3; i++) { 
    for (int j=0; j<2; j++) { 
      if (onMatrix[i][j] == hour && savedState[i] < 1) { 
        Serial.print("Switch on: "); 
        Serial.println(i);  
        mySwitch.switchOn(h, i+1); 
        savedState[i] = 1;  
        delay(500); 
      } 
      if (offMatrix[i][j] == hour && savedState[i] > 0) { 
        Serial.print("Switch off: "); 
        Serial.println(i);  
        mySwitch.switchOff(h, i+1); 
        savedState[i] = 0;  
        delay(500); 
      } 
    } 
  } 
} 
 
void print2digits(int number) { 
  if (number >= 0 && number < 10) { 
    Serial.write('0'); 
  } 
  Serial.print(number); 
} 
 
void setup() { 
  Serial.begin(9600); 
  mySwitch.enableTransmit(10); 
  delay(5000); 
  tmElements_t tm; 
  if (RTC.read(tm)) { setStartState(tm.Hour); }  
} 
 
void loop() { 
  tmElements_t tm; 
  if (RTC.read(tm)) { 
    print2digits(tm.Hour); Serial.write(':'); 
    print2digits(tm.Minute); Serial.write(':'); 
    print2digits(tm.Second); Serial.print(", Datum: "); 
    Serial.print(tm.Day); Serial.write('.'); 
    Serial.print(tm.Month); Serial.write('.'); 
    Serial.print(tmYearToCalendar(tm.Year)); 
    Serial.println(); 
    traverseMatrix(tm.Hour);  

  }  
  delay(50000); 
} 

Viele Wünsche lässt diese Schaltuhr nicht offen: Der häufigste Wunsch dürfte der nach 
mehr Schaltzeiten sein. Der ist einfach durch Vergrößerung der Arrays in zweiter Ebene 
zu erfüllen. Werden nicht alle Schaltzeiten für alle Verbraucher benötigt, setzen Sie zwei 
Schaltpunkte zur gleichen Uhrzeit, der zweite wird ignoriert. Oder setzen Sie zwei 
Ausschaltpunkte im Stundenabstand: Angenommen, die Wohnzimmerlampe wird um 
22:00 automatisch abgeschaltet, Sie greifen aber zur Fernbedienung, um noch etwas zu 
lesen, und vergessen schließlich um 22:20, die Lampe auszuschalten. In solch einem Fall 
ist ein zweiter Ausschaltpunkt um 23:00 ganz praktisch. Ein zweiter Wunsch dürfte der 
nach einem feineren Raster als Stunden sein: Fügen Sie einfach statt Stundenmarken 
eine dritte Array-Ebene ein, die Paare aus Stunden und Minuten enthält, und übergeben 
Sie den beiden Funktionen traverseMatrix() und setStartState() zwei Parameter. In 
der Funktion setStartState() müssen Sie darüber hinaus in einer weiteren 
Schleifenebene in jeder Stunde über die Minuten iterieren, um sicherzustellen, dass zwei 
Schaltpunkte innerhalb einer Stunde richtig erfasst werden.  

Zeitsignal: DCF77 

Das Schöne am DCF77-Zeitsignal des Mainflinger Senders ist, dass keine Uhr und 
kein Datum aufwendig gesetzt werden muss. Es muss keine Zeitzone mit passender 
Sommer- und Winterzeit eingestellt werden – alle Informationen werden bereits mit 
dem Funksignal übertragen. Doch es gibt auch Nachteile: Während eine RTC sofort 
nach dem Anschluss ausgelesen werden kann, dauert es bei einer Funkuhr zwei bis 
fünf Minuten, bis die Synchronisation erfolgt ist und Sie mit Zeit und Datum arbeiten 
können. Zudem gibt es Situationen, in denen die Funkabdeckung schlecht ist – häufig 
in verwinkelten Gebäuden, natürlich im Keller und dort, wo das Rauschen anderer 
Funksender oder schlecht geschirmte Schwingkreise stören. Des Weiteren ist das 
Funkmodul tendenziell teurer als eine RTC (unser billigstes DCF77-Modul war mit 5 
Euro doppelt so teuer wie unser günstigstes RTC-Modul).  

 

Bild 3.6: DCF77-Empfängermodul von Conrad: Die Anschlüsse im Abstand von 5,08 mm 
lassen sich zur Verwendung mit dem Breadboard mit Headern versehen. 

Zunächst gilt es, das DCF77-Modul und dessen Ausrichtung zu testen, praktisch 
hierfür ist der Beispiel-Sketch von Thijs Elenbaart, der im Abstand von 10 Milli-
sekunden prüft, ob der DCF77-Empfänger gerade eine 1 oder eine 0 liefert und diese 
auf dem seriellen Port ausgibt. Beim Wechsel von 0 auf 1 erfolgt ein Zeilenumbruch. 
Wirft man einen Blick auf die Signalcodierung, sollten folglich 58 Mal Signale von 
insgesamt einer Sekunde Länge und dann das zweisekündige Signal zur Minuten-
synchronisation folgen. Der High-Pegel des Sekundensignals sollte dabei 100 oder 
200 Millisekunden lang sein: Die Unterscheidung der beiden Signallängen stellt pro 
Minute maximal 59 Bit für Informationen bereit und überträgt so neben Datum, 
Stunde und Minute auch »Rufbits« für Mitarbeiter der Physikalisch-Technischen Bun-
desanstalt, Wetterinformationen und Informationen für den Katastrophenschutz.  

#define DCF77PIN 2 
 
int prevSensorValue=0; 
   
void setup() { 
  Serial.begin(9600); 
  pinMode(DCF77PIN, INPUT); 
  Serial.println("0ms.......100ms.....200ms.....300ms....."); 
} 
 
void loop() { 
  int sensorValue = digitalRead(2); 
  if (sensorValue==1 && prevSensorValue==0) {  
    Serial.println(""); 
  } 
  Serial.print(sensorValue);    
  prevSensorValue = sensorValue; 
  delay(10); 
} 

Schließen Sie den DCF77-Empfänger dann wie in der Skizze gezeigt an: »DCF output« 
des Empfängers an Digital 2 des Arduino. Digital 2 und 5 V werden zudem durch 
einen 10-kΩ-Pull-up-Widerstand verbunden. Ein paar Sekunden nach dem Sketch-
Upload und dem Start des Arduino sollten im Sekundentakt neue Zeilen erscheinen, 
die insgesamt 100±5 Zeichen lang sind und am Anfang eine Sequenz von 10 oder 
20±3 Einsen enthalten. Ist die Ausgabe deutlich weniger regelmäßig, ist die häufigste 
Ursache die Störung durchs PC-Netzteil. Gelegentlich hilft der Anschluss an ein gutes 
geregeltes separates Netzteil – in unserer Testumgebung führte erst der Anschluss 
ans Notebook zum Erfolg.  

Für die Auswertung des DCF77-Zeitsignals steht die ebenfalls von Thijs Elenbaart 
entwickelte DCF77-Bibliothek zur Verfügung. Diese interagiert mit der Time-Biblio-
thek und kann so als eine von vielen Zeitquellen für Arduino genutzt werden. Nach 
dem Import der DCF77-Bibliothek finden Sie unter Sketch → Beispiele → DCF77 → 
DCFbinaryStream ein hilfreiches Beispiel zur Auswertung des Funksignals und zum 
Debugging. Laden Sie diesen Sketch auf den Arduino und öffnen Sie die serielle Kon-

sole: Nach frühestens zwei Minuten sollte das korrekte Datum und die richtige Uhrzeit 
angezeigt werden. 

 

Bild 3.7: Anschluss des DCF77-Empfängers an den Arduino: Pull-up nicht vergessen. Masse 
und Versorgungsspannung sind bei unbeschrifteten Empfängern daran erkennbar, dass diese 
beiden Pins durch einen Kondensator verbunden sind. 

 

Bild 3.8: Sieht das Signal so aus, stimmen Stromversorgung und Ausrichtung der Antenne, jetzt 
können Sie das Zeitsignal auswerten. 

Hat der Erstkontakt mit der DCF77-Bibliothek geklappt, können Sie den Beispielcode 
mit RTC entsprechend anpassen. Dabei ist eine Besonderheit zu beachten: Da auf-
grund schlechten Empfangs die Funktion DCF.getTime() häufiger keine gültigen 
Werte zurückliefert, sollten Sie den DCF77-Empfänger dazu nutzen, die interne Uhr 
des Arduino zu setzen. Diese arbeitet zwar herzlich ungenau und benutzt nur die 
Funktion millis(), die die Millisekunden seit dem Start des Arduino zählt, ohne dabei 
Interrupts und andere Widrigkeiten zu berücksichtigen, doch das genügt, um mal eine 
Viertelstunde ohne DCF77-Empfang zu überstehen. Auf die Schaltskizze verzichten 
wir dieses Mal: Verbinden Sie den DCF77 wie oben gezeigt, der RF-Link-Sender wird, 
wie im Beispiel mit RTC gezeigt, mit Pin D10 verbunden. Die augenfälligste Änderung 
gegenüber dem RTC-Beispiel ist die Funktion setup(). Während wir bei einer per I²C 
angebundenen RTC immer davon ausgehen können, dass diese nach einigen Sekun-
den verfügbar ist, kann es bei einer Funkuhr mitunter fünf oder mehr Minuten dauern, 
bis nicht nur die Minutensynchronisation erfolgt ist, sondern auch alle Informations-
bits zum Datum richtig übertragen wurden. So lange verharren wir in einer while()-
Schleife. Im loop() wird bei Vorhandensein einer frischen DCF77-Information die 
interne Uhr neu synchronisiert und anschließend 50 Sekunden gewartet. Liegt keine 
frische DCF77-Information vor, prüfen wir bereits nach einer Sekunde erneut.  

Eine auf den ersten Blick kleine Änderung ist der doppelte Aufruf von switchOn() res-
pektive switchOff() im Abstand von 250 ms. Tatsächlich ist sie die Folge einer gra-
vierenden Änderung im Programmablauf. Ursache ist die Einbindung der DCF77-Bibli-
othek. Anders als die RTC-Bibliothek, die per I²C die Uhr pollt, wartet DCF77 nämlich 
auf Interrupt 0 auf eine steigende Flanke und macht dann genau das, was Interrupt 
wörtlich bedeutet: den Programmablauf unterbrechen und die Interrupt-Funktion 
abarbeiten. Ist bei eingehendem Sekundensignal des DCF77-Empfängers gerade ein 
Sendevorgang zur Funksteckdose aktiv, wird dieser für 100 oder 200 Millisekunden 
unterbrochen – viel zu lange bei etwa einer Millisekunde pro übertragenem Bit für die 
Kommunikation mit der Funksteckdose.  

#include <DCF77.h> 
#include <Time.h> 
#include <Wire.h> 
#include <RCSwitch.h> 
 
#define DCF_PIN 2 
#define DCF_INTERRUPT 0  
 
const int h = 1; 
RCSwitch mySwitch = RCSwitch(); 
DCF77 DCF = DCF77(DCF_PIN,DCF_INTERRUPT); 
int onMatrix[3][2] = { 
  { 10, 17 }, 
  { 11, 18 }, 
  { 10, 11 } 
}; 
int offMatrix[3][2] = { 

  { 11, 22 }, 
  { 12, 23 }, 
  { 3, 19 } 
}; 
int savedState[3] = { 0, 0, 0 }; 
int startState[3] = { 0, 0, 0 }; 
 
void setStartState(int hour) { 
  for (int k=0; k<=24+hour; k++) { 
    for (int i=0; i<3; i++) { 
      for (int j=0; j<2; j++) { 
        if (onMatrix[i][j] == k % 24) { 
          startState[i] = 1;  
        } 
        if (offMatrix[i][j] == k % 24) { 
          startState[i] = 0;  
        } 
      } 
    } 
  } 
  for (int i=0; i<3; i++) { 
     if (startState[i] > 0) { mySwitch.switchOn(h, i+1); } 
     if (startState[i] < 1) { mySwitch.switchOff(h, i+1); } 
     delay(100);  
  } 
} 
 
void traverseMatrix(int hour) { 
  for (int i=0; i<3; i++) { 
    for (int j=0; j<2; j++) { 
      if (onMatrix[i][j] == hour && savedState[i] < 1) { 
        Serial.print("Switch on: "); 
        Serial.println(i);  
        mySwitch.switchOn(h, i+1); 
        delay(250); 
 mySwitch.switchOn(h, i+1); 
 savedState[i] = 1;  
        delay(250); 
      } 
      if (offMatrix[i][j] == hour && savedState[i] > 0) { 
        Serial.print("Switch off: "); 
        Serial.println(i);  
        mySwitch.switchOff(h, i+1); 
 delay(250); 
 mySwitch.switchOff(h, i+1); 
        savedState[i] = 0;  
        delay(250); 

      } 
    } 
  } 
} 
 
void setup() { 
  Serial.begin(9600); 
  mySwitch.enableTransmit(10); 
  DCF.Start(); 
  delay(5000); 
  tmElements_t tm; 
  time_t DCFtime = DCF.getTime(); 
  while (DCFtime == 0) { 
    Serial.println("Warte auf DCF77..."); 
    delay(1000);  
    DCFtime = DCF.getTime(); 
  } 
  setTime(DCFtime); 
  setStartState(hour()); 
} 
 
void loop() { 
  print2digits(hour()); Serial.write(':'); 
  print2digits(minute()); Serial.write(':'); 
  print2digits(second()); Serial.print(", Datum: "); 
  Serial.print(day()); Serial.write('.'); 
  Serial.print(month()); Serial.write('.'); 
  Serial.print(year()); 
  Serial.println(); 
  traverseMatrix(hour());  
  time_t DCFtime = DCF.getTime(); 
  if (DCFtime!=0) { 
    setTime(DCFtime); 
    delay(50000); 
  } else { 
    delay(1000);   
  } 
} 
 
void print2digits(int number) { 
  if (number >= 0 && number < 10) { 
    Serial.write('0'); 
  } 
  Serial.print(number); 
} 

 



 

Funkuhr setzt RTC 

Sie haben zu den relativ präzisen DS3231-RTC-Bausteinen gegriffen und möchten diese 
nun möglichst genau setzen? Falls Sie daneben ein DCF77-Empfängermodul erworben 
haben, liegt es nahe, beide zu kombinieren: Solch eine Stand-alone-Lösung hat den Vorteil, 
dass Sie ein RTC-Modul – beispielsweise in einem Sensor im Garten – neu stellen können, 
ohne dass ein PC benötigt wird. Ein Arduino, ein 9-V-Block (oder ein Akkupack mit 
4 × 1,2 V) und ein Breadboard oder eine vorbereitete Platine genügen. Daneben werden 
lediglich die üblichen zwei bis fünf Minuten Geduld benötigt, bis das DCF77-Signal richtig 
eingefangen wurde. Die Verdrahtung nutzt wieder Pin D2 (Interrupt 0) für die Funkuhr 
sowie A4 und A5 (die bekannten I²C-Pins) für den Anschluss der RTC.  

Keine Kollisionen bei der Pin-
Belegung: Beide Zeitquellen aus den 
letzten Teilprojekten kombiniert, 
ergeben eine ultragenau 
synchronisierte RTC. 

Beim Sketch haben wir darauf 
geachtet, dass der Arduino auch 
ohne angeschlossenen PC nutzbar 
ist: So versetzen wir die LED an 
Pin D13 in ein 45 Sekunden langes 
Dauerleuchten, sobald ein ver-
wertbares Zeitsignal empfangen 
wurde und die RTC auf das 
ermittelte Datum gesetzt werden 
konnte. Zur Hilfe kommt dabei die 
Time-Bibliothek, die sowohl von 
DCF77 als auch von DS1307RTC 
verwendet wird – beide nutzen 
dieselbe Klasse, um Zeit/Datum-
Informationen zu speichern. Das 
erleichtert das Auslesen des 
DCF77-Signals und das Setzen 
der RTC mit diesem Wert. Konnte 
kein Signal empfangen werden 
oder tritt der unwahrscheinliche 
Fall ein, dass die RTC nicht justiert 
werden konnte, blinkt die LED an 
D13 im Viertelsekundentakt. So ist 
sichergestellt, dass die maximale 

Abweichung der RTC gegenüber der Atomuhr 0,5 Sekunden beträgt – viel genauer geht 
es bei dieser Art der Synchronisation nicht. Immerhin haben Sie so einen guten Startwert 
über das nächste Jahr: Ist dieses abgelaufen, sind auch bei DS3231 rund zwei Minuten 



 
 

Abweichung zu erwarten, und ein Neustellen der Uhr ist empfohlen. Nach dem Upload des 
Sketch verwenden Sie das Beispiel Ds1307ReadTest, um sich vom Erfolg zu überzeugen.  

#include <DCF77.h> 
#include <DS1307RTC.h> 
#include <Time.h> 
#include <Wire.h> 
 
#define DCF_PIN 2 
#define DCF_INTERRUPT 0  
 
time_t time; 
DCF77 DCF = DCF77(DCF_PIN,DCF_INTERRUPT); 
int ledPin = 13; 
 
void setup() { 
 pinMode(ledPin, OUTPUT); 
 Serial.begin(9600);  
 DCF.Start(); 
} 
 
void loop() { 
  time_t DCFtime = DCF.getTime(); 
  if (DCFtime!=0) { 
    tmElements_t tm;  
    breakTime(DCFtime, tm); 
    if (RTC.write(tm)) { 
      Serial.println("RTC erfolgreich gesetzt!"); 
      digitalClockDisplay(DCFtime); 
      digitalWrite(ledPin, HIGH); 
      delay(45000); 
      digitalWrite(ledPin, LOW); 
    } else { 
      delay(250);  
    }  
  } else { 
    digitalWrite(ledPin, HIGH); 
    delay(250); 
    digitalWrite(ledPin, LOW);   
    delay(250); 
  } 
} 
 
// Hilfsfunktion für serielle Ausgabe 
void digitalClockDisplay(time_t _time){ 
  tmElements_t tm;    
  breakTime(_time, tm); 
  Serial.println(""); 
  Serial.print("Zeit: "); Serial.print(tm.Hour); 
  Serial.print(":"); printDigits(tm.Minute); 
  Serial.print(":"); printDigits(tm.Second); 
  Serial.print(" Datum: "); Serial.print(tm.Day); 
  Serial.print("."); Serial.print(tm.Month); 
  Serial.print("."); Serial.println(tm.Year+1970); 
} 
 
// Hilfsfunktion für Formatierung 
void printDigits(int digits){ 
  Serial.print(":"); 
  if(digits < 10) 
    Serial.print('0'); 
  Serial.print(digits); 
} 

Projekt: Signale der China-434-Fernbedienung verarbeiten 
Jetzt zeigen wir Ihnen, wie Sie die Signale einer 434-MHz-Fernbedienung auslesen 
und so lernen, auch andere Geräte als Funksteckdosen anzusteuern. Zudem hilft das 
Gelernte, Signale von 434-MHz-Türklingeln oder -Wetterstationen auszuwerten.  

Benötigte Bauteile 

] Arduino Uno oder Pro Mini 

] RF-Link-Empfänger, meist Bestandteil eines RF-Link-Kits (z. B. Watterott WRL-
10532), 2 bis 6 Euro 

Benötigte Software 

] RCSwitch-Bibliothek – getestet mit Version 2.51 (code.google.com/p/rc-switch/) 

 

Bild 3.9: Fernbedienung der China-Steckdosen: Ein simpler IC erzeugt das wenige Bit lange 
Funksignal. 

Wenn Sie sich näher mit den günstigen Funkaktoren (Steckdosen, Lampenfassungen, 
Unterputzdosen) auf 434 MHz auseinandersetzen, werden Sie früher oder später 
darauf stoßen, dass der im Sender verwendete IC keinesfalls fixe 5 Bit für den 
Hauscode und weitere 5 für die Geräteadresse verwendet. Tatsächlich sind nur die 
ersten sechs Pins des SC5262 Adresspins, alle weiteren können flexibel genutzt wer-
den. Eine Besonderheit der Adresspins ist die Ansteuerung als »Tristate-Pins«. Sie 
erinnern sich sicher an die Arduino-Grundlagentutorials, die erklären, dass ein digitaler 
Eingang immer einen definierten Zustand haben muss: Dafür kommt in der Regel ein 
Pull-up- oder Pull-down-Widerstand zum Einsatz, der dafür sorgt, dass ein Pin garan-
tiert auf Spannung oder garantiert auf Masse liegt. Lässt man diesen Widerstand weg, 
erhält man einen – bei Digitalelektronik unerwünschten – dritten Zustand. Diesen 

macht sich der Tristate-Encoder zunutze und liest ihn aus. Statt 1.024 (210) möglicher 
Adressen ergeben sich in Trinärlogik satte 310, also 59.049 Adressen. Zwar verwen-
den die meisten Hersteller binäre Dip-Switches und bilden diese in der Regel auf 0 und 
den dritten Zustand F (Floating) ab, um beim Senden etwas Energie zu sparen, verein-
zelt sind jedoch Geräte mit echter trinärer Adressierung im Handel und solche, bei 
denen nicht jeder (binäre) Dip-Schalter auf 0 und F verdrahtet ist. Auch ist es nicht 
immer klar, welche Bits ein Gerät zur Adressierung verwendet und wie die Datenbits 
aussehen. In solch einem Fall hilft die »Aufnahmefunktion« der Bibliothek RCSwitch.  

Unser Sketch basiert auf dem Beispiel-Sketch ReceiveDemo_Advanced der RCSwitch-
Bibliothek. Eingefügt haben wir eine Bereitschaftsanzeige (»Bitte Taste auf Fernbedie-
nung drücken...«) und die Ausgabe der »Uptime« des Arduino. Die dafür verwendete 
Funktion millis() zählt die Millisekunden seit dem Boot des Mikrocontrollers, nutzt 
dafür aber den etwas unpräzisen externen Quarz und berücksichtigt keine Interrupts. 
Innerhalb von setup() sorgt mySwitch.enableReceive() dafür, dass der übergebene 
Interrupt (Int 0 ist Pin D2, Int 1 ist Pin D3) zugewiesen wird. Innerhalb des loop() wird 
ohne Pause geprüft, ob mySwitch() noch nicht verarbeitete Daten enthält. Ist das der 
Fall, bereitet sie die Funktion output() hübsch auf und gibt sie über die serielle Ver-
bindung aus. 

#include <RCSwitch.h> 
 
RCSwitch mySwitch = RCSwitch(); 
 
void setup() { 
  Serial.begin(9600); 
  Serial.println("Bitte Taste auf Fernbedienung druecken...");  
  mySwitch.enableReceive(0);  // Int 0 = Pin 2 
} 
 
void loop() { 
  if (mySwitch.available()) { 
    Serial.print("Uptime: "); 
    Serial.println(millis()); 
    output(mySwitch.getReceivedValue(),  
      mySwitch.getReceivedBitlength(),  
      mySwitch.getReceivedDelay(),  
      mySwitch.getReceivedRawdata(), 
      mySwitch.getReceivedProtocol()); 
    mySwitch.resetAvailable(); 
  } 
} 

Bild 3.10: Einfachster Anschluss: DATA des Receivers wird mit Pin D2 (Interrupt 0) des 
Arduino (hier Wattuino Pro Mini) verbunden, VCC mit 3,3 oder 5 V, GND mit Masse. 

 

Bild 3.11: Das Beispiel zum Empfangen von Fernbedienungssignalen hilft bei der Fehlersuche: 
Welche Codes sendet eine Fernbedienung aus? Wie lang sind die Pulse? 

Wir haben nun die Signale verschiedener Fernbedienungen und die mit der RCSwitch-
Bibliothek erzeugten Signale aufgefangen und ausgewertet: Auffällig sind die abwei-
chenden Pulslängen, die zwischen 340 und 430 Mikrosekunden lagen. Probleme bei 

der Steuerung können durchaus hierin begründet liegen – als Gegenmittel bietet die 
Bibliothek RCSwitch das Setzen der Pulslänge auf einen von 350 Mikrosekunden 
abweichenden Wert. Erkennen können Sie mit diesem Beispiel auch Funksteckdosen-
systeme, die ein abweichendes Codierungsschema verwenden, so sind Gerätefamilien 
am Markt, die vier Dip-Schalter für den Hauscode verwenden und sechs zur Geräte-
adressierung – die gesamte Adresse ist immer noch zehn Tristate-Bits lang, aber das 
fünfte Bit ändert sich beim Ansteuern verschiedener Steckdosen.  

Der RF-Link-Empfänger54 ist ein simpelstes Stück Hardware, das alles empfängt, was 
gerade auf 433,92 MHz gesendet wird. Die Auswertung des Funksignals, eine Integ-
ritätsprüfung und die Adresszuordnung muss in diesem Fall der angeschlossene 
Mikrocontroller erledigen. Da dessen Arbeit durch ständige Interrupts des RF-Link-
Empfängers unterbrochen wird, ist der Dauereinsatz eher nicht zu empfehlen: Zum 
einen wird einer von nur zwei Hardware-Interrupt-Pins belegt, zum anderen wird der 
Arduino dadurch in Umgebungen mit vielen 434-MHz-Sendern extrem langsam. 
Sollten Sie auf rohe, über 434 MHz übertragene Daten angewiesen sein, ist es sinn-
voller, einen Arduino Pro Mini als I²C-Slave (oder -Multimaster) anzubinden und 
diesen ausschließlich mit der Verarbeitung der 434-MHz-Funksignale zu betrauen.  

Projekt: FS20-Aktoren steuern 
Sie haben bereits FS20-Aktoren wie Funksteckdosen oder Dimmer im Einsatz oder 
planen die Anschaffung von Funkheizkörperthermostaten mit separatem Raumther-
mostat? Dann sollten Sie sich mit dem Anschluss eines FS20-US-Senders an den 
Arduino vertraut machen.  

Benötigte Bauteile 

] Arduino Uno oder Pro Mini 

] FS20-US-Sendemodul, ca. 20 Euro 

] FS20-Funksteckdose oder -dimmer, neu ca. 25 bis 30 Euro, gebraucht bei eBay ab 
15 Euro 

] eine FS20-Fernbedienung ist nicht notwendig 

Benötigte Software 

] keine, die programmieren wir selbst 

Unter dem Namen »FS20 US« (Universalsender) vertreibt ELV (und Conrad sowie 
andere Elektronikversender) ein Sendemodul für das FS20-System. Die rund 20 Euro 
teure Platine stellt eine serielle Schnittstelle bereit, über die Steuerungsbefehle zu 
beliebigen FS20-Aktoren geschickt werden können. Recht ungewöhnlich ist der Auf-
bau der Platine: Diese ist als Sandwich ausgeführt, Sie müssen zwei Platinen mit 
doppelseitigem Klebeband mechanisch verbinden und dann auf vier SMD-Lötpads 
den Kontakt zwischen den Sandwich-Hälften sicherstellen. Das klingt schwieriger, als 
es ist, denn die SMD-Lötpads sind im Zehntel-Zoll-Raster angeordnet und lassen sich 

mit ein wenig Flussmittelpaste leicht verlöten. Anschließend müssen Sie sich noch 
entscheiden, ob Sie gerade oder gewinkelte Kontaktstifte anbringen. Aufgrund der in 
die Platine integrierten Antenne nehmen Sie damit die Entscheidung von Aufbau und 
Positionierung der gesamten Sendeeinheit teilweise vorweg. Schmökern Sie ein wenig 
im beiliegenden Handbuch, dieses erklärt die Details der FS20-Kommunikation viel 
besser, als es der Platz in diesem Buch zulässt.  

 

Bild 3.12: Löten erforderlich: Die Platine ist als Sandwich aus zwei Platinenteilen verklebt, 
Lötbrücken müssen Sie selbst ziehen – das ist nicht schwieriger, als Sockelleisten anzulöten. 

Im Gegensatz zu den simplen RF-Link-Sendern für wenige Euro integriert der Univer-
salsender für FS20 die gesamte Sendelogik. So sorgt ein eigener Mikrocontroller dafür, 
dass einige Einstellungen nicht-flüchtig auf dem Modul gespeichert werden. Und dank 
der seriellen Schnittstelle müssen Sie nur ein paar Bytes an Senderbefehlen, Hauscode 
und Gerätecode sowie Steuerbefehlen zum FS20 US schicken – für die Modulation des 
Signals sorgt in diesem Fall der FS20 US. Bleibt ein kleines Problem: Die serielle 
Schnittstelle des Arduino wird im Regelfall für Debug-Ausgaben mit dem seriellen 
Monitor genutzt. Beim Anschluss des Moduls ist darauf zu achten, dass Rx mit dem 
Tx-Pin des Arduino verbunden wird und Tx des Senders mit Rx des Arduino. Unser 
Sketch verwendet Arduinos Pin D12 als Rx und D13 als Tx. »Transmission« auf 13 zu 
legen, hat den Vorteil, dass Sie es (mit ein wenig Glück und kurzem Abstand) sehen 
können, wenn Daten zum Sendemodul übertragen werden.  

Bild 3.13: Achten Sie beim Anschluss serieller Verbindungen darauf, dass der Tx-Pin des einen 
Geräts jeweils mit dem Rx-Pin des anderen verbunden ist. Als Versorgungsspannung verträgt 
der FS20 US 3,3 V bis 5 V – läuft also mit allen Arduinos. 

Den Erstkontakt mit dem FS20 US haben wir bewusst einfach gehalten: Wir initiali-
sieren die softwareemulierte serielle Schnittstelle auf den Pins D12 und D13 und 
schicken dem FS20 US ständig die Bytefolge 0x02, 0x01, 0xF0. Das erste Byte leitet 
einen neuen Befehl ein, das zweite Byte teilt die Länge des folgenden Befehls mit (ein 
Byte), und das dritte Byte fragt Status und Baudrate an. Bytes für das Ende eines 
Befehls kennt der FS20 US nicht, stattdessen genügt es, etwas mehr als 10 ms zu 
warten. In einer while()-Schleife lesen wir nun die zurückgegebenen Bytes aus und 
geben sie über die serielle Konsole aus. Die vom FS20 zurückgegebene Nachricht 
sollte 2201 lauten. 0x02 leitet die Datensequenz ein, 2 bedeutet, es folgen 2 Byte an 
Information: 0 steht für »Stand-by« und 1 für die eingestellte Baudrate von 9600 
(FS20-US-Handbuch, Seite 7). Wollen Sie die Begrüßung des FS20 US (»ELV FS20 
US – vX.Y.Z«) zu Gesicht bekommen, müssen Sie etwas tricksen und den FS20 US 
nicht über die 5-V-Schiene, sondern über einen erst nach dem Start der seriellen 
Verbindung aktiven digitalen Pin mit Spannung versorgen.  

#include <SoftwareSerial.h> 
#define rxPin 12 
#define txPin 13 
 
SoftwareSerial softSerial =  SoftwareSerial(rxPin, txPin); 
 
void setup() { 
  softSerial.begin(9600); 
  Serial.begin(9600); 
  Serial.println("Arduino ist bereit"); 
} 
 
void loop() { 
  softSerial.write(0x02); // Präfix   
  softSerial.write(0x01); // Befehlslänge 
  softSerial.write(0xF0); // Befehl: Status 
  delay(15); 
  while (softSerial.available()) { 
    int b;  
    b = softSerial.read(); 
    Serial.print(b); 
  } 
  Serial.println(); 
  delay(1000); 
} 

Wir wissen nun, dass der Sender ansprechbar ist, und können versuchen, Steckdosen 
zu schalten. Uns lag dafür eine jungfräuliche Steckdose vor, was bedeutet, wir konnten 
Geräteadresse und Hauscode frei wählen. Den Hauscode setzten wir auf 0xCA, 0xFE, 
als erste Geräteadresse nutzten wir 0x00 (da bleibt Luft nach oben). Wenn Sie bereits 
FS20-Geräte im Einsatz haben, müssen Sie ein wenig rechnen: FS20-Adressen wer-
den im Quaternärsystem (Vierersystem) abgebildet, aber perfiderweise nicht mit 0 
bis 3 repräsentiert, sondern mit 1 bis 4. Ziehen Sie also von jeder Stelle 1 ab. Aus dem 
Gerätecode 1243 wird so 0132. Rechnen Sie nun jeden quartenären Zweierblock in 
eine einzige hexadezimale Ziffer um: 32 zur Basis 4 entspricht 3·4 + 2·1 = 14 zur Basis 
10. Das wiederum entspricht 0xE zur Basis 16. 01 zur Basis 4 ist einfacher umzurech-
nen – es entspricht 0x1 hexadezimal. Aus der FS20-Adresse 1243 wird somit 0x1E. 

Unser simpler Sketch schickt im Abstand von zwei Sekunden den Toggle-Befehl, bei 
Lichtschaltern und Funksteckdosen schaltet dieser zwischen »An« und »Aus« um. 
Aus Gründen der Übersichtlichkeit haben wir ihn in eine eigene Funktion ausgelagert. 
Wenn Ihre Funksteckdosen bereits auf die vergebenen Codes angelernt sind, testen 
Sie so, ob Sie die Adressierung korrekt vorgenommen haben. Eine neue Funksteckdose 
lernen Sie an, indem Sie nach dem Start des Sketch einige Sekunden die Taste an der 
Steckdose drücken, bis diese zu blinken beginnt. Dann lassen Sie die Taste los, wenige 
Sekunden später ist die Steckdose angelernt und blinkt im Abstand von drei Sekunden. 
Als Antwort sollte wieder 2201 ausgegeben werden. Lautet die Antwort 2211, ist das 

Modul noch mit dem Senden des letzten Befehls beschäftigt. Bei 2251 haben Sie zu 
viel gesendet und Ihren »Duty Cycle« aufgebraucht – das 868,35-MHz-Band darf 
jeder Sender nur 1 % der Zeit nutzen, im Fall des FS20 US kümmert sich das Sende-
modul selbst um die Einhaltung dieser regulatorischen Vorgabe.  

Um weitere Steckdosen anzulernen, zählen Sie die Geräteadresse, in der Funktion 
fs20toggle() einfach hoch und führen die letzten Schritte erneut durch. Werfen Sie 
auch einen Blick ins Handbuch: Da toggle zwar zum Testen schön ist, aber wenig 
praktisch, wenn es darum geht, einen definierten Zustand herzustellen, nutzen Sie die 
Befehle 0x00 (Aus), 0x10 (Einschalten auf 100 %) und 0x11 (Einschalten auf letzter 
Helligkeitsstufe) – 0x01 bis 0x0f kennzeichnen die verschiedenen Helligkeitsstufen bei 
Dimmern.  

#include <SoftwareSerial.h> 
#define rxPin 12 
#define txPin 13 
#define houseCode1 0xCA 
#define houseCode2 0xFE 
 
SoftwareSerial softSerial =  SoftwareSerial(rxPin, txPin); 
 
void fs20toggle(int actuator) { 
  softSerial.write(0x06); // Befehlslänge 
  softSerial.write(0x02); // Präfix   
  softSerial.write(0xF1); // einmal schalten 
  softSerial.write(houseCode1); 
  softSerial.write(houseCode2); 
  softSerial.write(actuator); 
  softSerial.write(0x12); // "toggle" 
  softSerial.write(0x01); // wird ignoriert 
  delay(50); 
  while (softSerial.available()) { 
    int b;  
    b = softSerial.read(); 
    Serial.print(b); 
  } 
  Serial.println(); 
} 
 
void setup() { 
  softSerial.begin(9600); 
  Serial.begin(9600); 
  Serial.println("Arduino ist bereit"); 
} 
 
void loop() { 
  fs20toggle(0x00); 
  delay(3000); 
} 

Heizkörperthermostat, Funksteckdose, SMS-Rauchmelder,
RFID-Katzentür, Gartenbewässerung & Co.: Die Aufgaben
im Haus übernimmt ab jetzt der ArduinoTM.

Smart Home ist in aller Munde, aber noch längst nicht in
allen Wohnungen und Häusern. Woran liegt es? Am feh-
lenden Wissen und an den Kosten. Fertige Lösungen sind
teuer, kompliziert und nicht flexibel genug. Wer möchte
täglich den Monteur im Haus haben, weil das Licht oder
die Heizung sich nicht mehr einschaltet? Bleiben Sie Herr
im Haus und automatisieren Sie selbst und das auch noch
für kleines Geld.

Für eigene Elektronikprojekte bietet sich der ArduinoTM geradezu an:
Günstig, flexibel und eine große Community, die auch bei Problemen
gerne und schnell hilft. Wenn es darum geht, den ArduinoTM im eigenen
Heim zu installieren, muss es auch nicht immer eine fertige Platine
sein, wie z. B. der ArduinoTM Uno. Der Mikrocontroller genügt auch.
Mattias Schlenker zeigt Ihnen, wie Sie für wenige Euro einen ArduinoTM

nachbauen. Neben dem ersparten Geld sparen Sie auch noch Platz. 
Um nicht so viel Strom zu verbrauchen, können Sie den ArduinoTM

auch schlafen legen. Wie das geht, erfahren Sie in diesem Buch.

Der erfahrene Autor zeigt Ihnen alles, was Sie zur Hausautomatisie-
rung benötigen. Angefangen von Hardware über Software bis zum
Entwurf von Schaltungen ist alles dabei. Dem Thema Energieversor-
gung der aufgebauten Schaltung ist ein eigenes Kapitel gewidmet.
Leser ohne ArduinoTM-Kenntnisse erhalten im Anhang einen Crashkurs
zur Programmierung. Neben Grundlagen zeigt Schlenker viele nützliche
Projekte: Funksteckdosen schalten, Sensoren nutzen, RFID-Katzentür
oder auch einen Lecksensor für die Waschküche.

Jedes Projekt startet mit einer Liste der benötigten Hard- und 
Software. Damit können Sie die Projekte direkt nachbauen und auch
erweitern. Da Maker nicht gerne wegwerfen, zeigt Ihnen Schlenker,
wie Sie Upcycling betreiben und alte Geräte für neue Dinge nutzen
können.

Der komplette Quellcode aus dem Buch auf www.buch.cd

Aus dem Inhalt:
• Freie Projekte zur Hausautomation

• Sauberer Aufbau von Schaltungen

• Upcycling: vorhandene Dinge nutzen

• Gehäuse bauen

• Tipps zur Energieversorgung

• ArduinoTM Yún als Steuerungszentrale

• Raspberry Pi als Alternative

• Arduinos für Sensoren und Aktoren

• ArduinoTM selber bauen

• Funksteckdosen schalten

• Licht- und Temperatursensoren

• Lecksensor für die Waschküche

• Fenster überwachen

• Steuerungszentrale aufbauen

• RFID-Katzentür

• Twitternde Zimmerpflanzen mit
Temboo

• Briefkasten überwachen

• Anwesenheitsermittlung per Bluetooth

• Rauchmelder meldet per SMS

• Gaslecksensor selbst gebaut

Besuchen Sie 
unsere Website
www.franzis.de

Über den Autor:
Mattias Schlenker entwickelt Live-Linux-
Distributionen vornehmlich als Notfall- und
Rettungssysteme und ist als Autor für Linux-,
Netzwerk- und Sicherheitsthemen für verschie-
dene Computermagazine tätig. Zu ArduinoTM

kam der gelernte Kraftfahrzeugmechaniker aus
Neugier, als er nach einer günstigen Möglich-
keit suchte, Betriebsdaten seines PKW über die
OBD-II-Schnittstelle auszulesen. Mittlerweile
ist er fasziniert von der Möglichkeit, mit 
ArduinoTM die verschiedensten Mess-, Regelungs-
und Steuerungsaufgaben einfach, günstig und
schnell umsetzen zu können – dementsprechend
viele Arduinos sind in seinem Haushalt im 
Einsatz. Das in den letzten Jahren neu erworbene
ArduinoTM-Fachwissen gibt er gerne in Buchform
weiter.

MATTIAS SCHLENKER

